
1. [bookmark: _Toc122434485][bookmark: _Toc51770678]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.2.2.3.1 which is part of the 5G ENDC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	Shaun.harry@keysight.com
	
0. [bookmark: _Toc51770679]Verification Test Summary 
Test Case: 	8.2.2.3.1
ATS Version:	iwd-TTCN3-B2019-06_D20wk25
[bookmark: _Hlk37919671]System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	MediaTek MT6297
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc51770680]Corrections required
[bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]
[bookmark: OLE_LINK58][bookmark: OLE_LINK57][bookmark: OLE_LINK15][bookmark: OLE_LINK12][bookmark: OLE_LINK9][bookmark: OLE_LINK8]2.1 Change 1
	Function name
	fl_TC_8_2_2_3_1_ENDC_EUTRA_TestBody


	[bookmark: _Hlk15652477][bookmark: _Hlk5709449]Reason for change
	1. As per test specification, Pdcp duplication shall be set to false in 8.2.2.3.1. common test step f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit() is used to execute step 1-2 of test cases 8.2.2.3.1 and 8.2.2.3.1. There is no parameter passed to this function to differentiate pdcp duplicate enabled or disabled scenarios.

1. SRB3 is configured with secondary Key(KgNB) at step #1, However SRB1/2 shall be configured with master key (KeNB). Hence the SRB3 message sent at step #3 should be send with KgNB keys. Master keys configured after Step #2 need to be reconfigured back to kgNB keys before Step#3 SRB3 message can be send out. 

1. There is no wait in EUTRA PTC before Step #5, This will lead to EUTRA proc advancing to Step #5, when NR proc is performing Step 3-4. This might cause a mismatch in security keys.

1. Step #5 SRB1 messageshall use master keys. Since secondary keys are configured at step #3 (See change 2), there is a need to configure back the KeNB keys before step #5.

1. NR proc shall send both RB and Reconfig Coordination messages at step #5. However RB is not handled in EUTRA Proc at step #5.

	Summary of change
	1. Called f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit()  with new parameter pdcp duplicate set to false.

1. Call to f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys() added before Step 3 with SK Counter value 1, since SK Counter value is update at step #1.

1. Added call to f_EUTRA_NR_WaitForCoOrd_Trigger() to wait for trigger from NR proc before starting Step #5.

1. Call to f_EUTRA38_ENDC_Send_eNBKeys() added before Step 5

1. Added wait for NR Cordination message “RB” at step 5 and also send the received RBConfig to function f_EUTRA38_508_BuildRRCConnReconfig_NR_Config_Setup() to be sent to device as part of DL message.

	TTCN module
	RRCReconfig_ENDC_EUTRA.ttcn

	MCC160 Comment
	



Before Change
[bookmark: OLE_LINK53][bookmark: OLE_LINK52]
function fl_TC_8_2_2_3_1_ENDC_EUTRA_TestBody() runs on EUTRA_5GS_PTC
  {// @sic R5-202647 sic@
    var template (omit) octetstring v_NR_SCGConfig := omit;
    var template (value) RRCConnectionReconfiguration_v890_IEs v_RRCConnectionReconfiguration_v890_IEs;
    var EUTRA38_SecurityParamsExtd_Type v_Security_Params := f_EUTRA38_Security_Get();
    
    // @siclog "Steps 1-2" siclog@
    f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit(eutra_Cell1, true, (v_Security_Params.SK_Counter + 1)); 
    
    // @siclog "Step 3" siclog@
    // The SS transmits on SRB3/SCG an NR RRCReconfiguration message to reconfigure SCG DRB.
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);

    // @siclog "Step 5" siclog@
    // The SS transmits over split SRB1 SCG path/NR Cell 1 an RRCConnectionReconfiguration message to release SRB3.
    v_NR_SCGConfig := f_EUTRA_NR_WaitForCoOrd_Data(NR, "Reconfig");
    v_RRCConnectionReconfiguration_v890_IEs := f_EUTRA38_508_BuildRRCConnReconfig_NR_Config_Setup (v_NR_SCGConfig); 
    
    SRB.send(cas_SRB1_RrcNasPduList_REQ(eutra_Cell1,
	                                        cs_TimingInfo_Now,
	                                        cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,-,-,-,-,v_RRCConnectionReconfiguration_v890_IEs),
	                                        omit));
	…

  }



After Change

function fl_TC_8_2_2_3_1_ENDC_EUTRA_TestBody() runs on EUTRA_5GS_PTC
  {// @sic R5-202647 sic@
    var template (omit) octetstring v_NR_SCGConfig := omit;
    var template (omit) octetstring v_NR_RBConfig := omit; 
    var template (value) RRCConnectionReconfiguration_v890_IEs v_RRCConnectionReconfiguration_v890_IEs;
    var EUTRA38_SecurityParamsExtd_Type v_Security_Params := f_EUTRA38_Security_Get();
    
    // @siclog "Steps 1-2" siclog@
    f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit(eutra_Cell1, true, (v_Security_Params.SK_Counter + 1), false); 
    
    // @siclog "Step 3" siclog@
    // The SS transmits on SRB3/SCG an NR RRCReconfiguration message to reconfigure SCG DRB.
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger);
    f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR, 1); 
    
    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);        
    f_EUTRA38_ENDC_Send_eNBKeys(NR);            
    // @siclog "Step 5" siclog@
    // The SS transmits over split SRB1 SCG path/NR Cell 1 an RRCConnectionReconfiguration message to release SRB3.
    v_NR_SCGConfig := f_EUTRA_NR_WaitForCoOrd_Data(NR, "Reconfig");
    v_NR_RBConfig := f_EUTRA_NR_WaitForCoOrd_Data(NR, "RB"); 
    v_RRCConnectionReconfiguration_v890_IEs := f_EUTRA38_508_BuildRRCConnReconfig_NR_Config_Setup (v_NR_SCGConfig, v_NR_RBConfig); 
    
    SRB.send(cas_SRB1_RrcNasPduList_REQ(eutra_Cell1,
	                                        cs_TimingInfo_Now,
	                                        cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,-,-,-,-,v_RRCConnectionReconfiguration_v890_IEs),
	                                        omit));
	  ….

  }



2.2 Change 2
	Function name
	f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit

	Reason for change
	1. As per test specification, Pdcp duplication shall be set to false in 8.2.2.3.1. common test step f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit() is used to execute step 1-2 of test cases 8.2.2.3.1 and 8.2.2.3.1. There is no parameter passed to this function to differentiate pdcp duplicate enabled or disabled scenarios.

1. At step 1 of the test case Key to use is updated as mater for the Split SRB1/SRB2. Hence Keys need to be updated to KeNB to handle the UL response message after sending the DL message to device. 


	Summary of change
	1. Added a new parameter to test step f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit()  to differentiate PDCp Enabled and PDCp disabled scenarios. This parameter shall be used to handle PDCP duplicate scenario and non duplicate scenarios to expect UL PDUs either in both RLC paths (EUTRA + NR) or only in one of the path.

1. Call to test step f_EUTRA38_ENDC_Send_eNBKeys() is added after step #1 to send KeNB Keys to NR proc.


	TTCN module
	RRCReconfig_ENDC_EUTRA.ttcn

	MCC160 Comment
	



Before Change

function f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit(EUTRA_CellId_Type   p_CellId,
                                                  boolean             p_Receive_RBConfig2 := false,
                                                  integer p_SK_Counter := 0
                                                  ) runs on EUTRA_5GS_PTC
  {
    …..
    
    f_EUTRA_508RRC_IntraLTE_HO_IntraCell_Step1_5(p_CellId,
                                                 cs_TimingInfo_EUTRA_NB(v_Timing),
                                                 v_ConfigDrbSS,
                                                 -,  //default value tsc_C_RNTI_Def2
                                                 -, //MeasConfig,
                                                 -,
                                                 v_SecurityActTimeList,
                                                 -,
                                                 v_RRCConnectionReconfiguration_v890_IEs,
                                                 v_SrbDrbList
                                                 );
    
    // Check: Does the UE send a RRCConnectionReconfigurationComplete message containing NR RRCReconfigurationComplete message on split SRB1 duplicated over the SCG path?
	interleave {
	        []SRB.receive(car_SRB_RrcPdu_IND_Common(eutra_Cell1,tsc_SRB1, cr_RRCConnectionReconfigurationComplete_r15, -, crs_RlcBearerRouting_Eutra(eutra_Cell1))) {}
	        []SRB.receive(car_SRB_RrcPdu_IND_Common(eutra_Cell1,tsc_SRB1, cr_RRCConnectionReconfigurationComplete_r15, -, cr_RlcBearerRouting_NR(nr_Cell1))) {}
	        
	}
    
    
    //After steps in table 8.2.2.2.1.3.2-2 are completed the SS is configured to send RLC ACK on split SRB1 and the last message received in step 2 is acknowledged.
    f_EUTRA_SS_CommonRadioBearerConfig(eutra_Cell1, {cs_RB_Reconfig_TestModeAckProhibit( tsc_RbId_SRB1, Continue)});

….                                                
    }



After Change

function f_EUTRA38_ReconfigureSRBs_NR_PDCPSplit(EUTRA_CellId_Type   p_CellId,
                                                  boolean             p_Receive_RBConfig2 := false,
                                                  integer p_SK_Counter := 0,
                                                  boolean             p_DuplicatePDU := true
                                                  ) runs on EUTRA_5GS_PTC
  {
   ….
    v_Timing :=  f_EUTRA_GetNextSendOccasion(p_CellId, tsc_EUTRA_DelayBeforeIntraCellHO);
    
    f_EUTRA_508RRC_IntraLTE_HO_IntraCell_Step1_5(p_CellId,
                                                 cs_TimingInfo_EUTRA_NB(v_Timing),
                                                 v_ConfigDrbSS,
                                                 -,  //default value tsc_C_RNTI_Def2
                                                 -, //MeasConfig,
                                                 -,
                                                 v_SecurityActTimeList,
                                                 -,
                                                 v_RRCConnectionReconfiguration_v890_IEs,
                                                 v_SrbDrbList
                                                 );
    f_EUTRA38_ENDC_Send_eNBKeys(NR);            
    if (p_DuplicatePDU)
    {
	    // Check: Does the UE send a RRCConnectionReconfigurationComplete message containing NR RRCReconfigurationComplete message on split SRB1 duplicated over the SCG path?
	    interleave {
	        []SRB.receive(car_SRB_RrcPdu_IND_Common(eutra_Cell1,tsc_SRB1, cr_RRCConnectionReconfigurationComplete_r15, -, crs_RlcBearerRouting_Eutra(eutra_Cell1))) {}
	        []SRB.receive(car_SRB_RrcPdu_IND_Common(eutra_Cell1,tsc_SRB1, cr_RRCConnectionReconfigurationComplete_r15, -, cr_RlcBearerRouting_NR(nr_Cell1))) {}
	        
	    }
    }
    else
    {
      // Check: Does the UE send a RRCConnectionReconfigurationComplete message containing NR RRCReconfigurationComplete message on split SRB1 on MCG or SCG?
	    alt {
	        []SRB.receive(car_SRB_RrcPdu_IND_Common(eutra_Cell1,tsc_SRB1, cr_RRCConnectionReconfigurationComplete_r15, -, crs_RlcBearerRouting_Eutra(eutra_Cell1))) {}
	        []SRB.receive(car_SRB_RrcPdu_IND_Common(eutra_Cell1,tsc_SRB1, cr_RRCConnectionReconfigurationComplete_r15, -, cr_RlcBearerRouting_NR(nr_Cell1))) {}
	        
	    }
    }
    //After steps in table 8.2.2.2.1.3.2-2 are completed the SS is configured to send RLC ACK on split SRB1 and the last message received in step 2 is acknowledged.
    f_EUTRA_SS_CommonRadioBearerConfig(eutra_Cell1, {cs_RB_Reconfig_TestModeAckProhibit( tsc_RbId_SRB1, Continue)});

…                                                
    }


2.3 Change 3
	Function name
	fl_TC_8_2_2_3_1_ENDC_NR_TestBody

	Reason for change
	1. As per test specification, Pdcp duplication shall be set to false in 8.2.2.3.1. 

1. As per section 5.3.1.2 of 38.331, “The integrity protection and ciphering algorithm is common for signalling radio bearers SRB1 and SRB2.”. Also, as per 36.331, E-UTRAN configures the same RAT type (i.e. EUTRA or NR) for PDCP configuration of SRB1and SRB. However, At step #1, only SRB1 is moved from EUTRA PDCP to NR PDCP, which is invalid configuration.  It is required to update  SRB2 as well along with SRB1.

1. Template cs_NR_RlcBearerConfig_AM is used to configure SS side SRB. This will end up using template to configure IE “sn_FieldLength” with value snSize18. For Signalling bearers sn_FieldLength shall be set as snSize12.

1. Template used to configure SRB3 at SS side cs_NR_SS_RadioBearer_Config , doesn’t contain RLC Configuration.

1. When f_NR_SendRRCReconfigurationContentsToEUTRA is called to send the v_RadioBearerConfig2 at step 1 to EUTRA, there is no need to send v_SCGConfig, since this information has already send to EUTRA Proc along with v_RadioBearerConfig in the earlier line.

	Summary of change
	1. Corrected parameter value pdcp_duplicate to false at step #1 when template cs_NR_PDCP_Config_SplitSRB is called.

1. Added code to configure SRB2 as split SRB along with SRB1.

Note : An associated prose change on 38.523-1 will be raised at RAN5#89

1. New Template cs_NR_RlcBearerConfig_AM_SRB is defined, which will set the sn_FieldLength as Size12 as required for SRBs. New template is used to configure SS at step #1.

1. Template cs_NR_SS_SRB3_Config is used to configure SS side SRB3, which has additional RLC configuration as well for the Bearer.

1. Removed  v_SCGConfig from the parameter list, when f_NR_SendRRCReconfigurationContentsToEUTRA is called secons time at step1 to send v_RadioBearerConfig2.

	TTCN module
	RRCReconfig_ENDC_NR.ttcn

	MCC160 Comment
	



Before Change

function fl_TC_8_2_2_3_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC
  {// @sic R5-202647 sic@
    var template (value) CellGroupConfig v_SCGConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig1;
    var template (value) RadioBearerConfig v_RadioBearerConfig2;
    var template (omit) octetstring v_EncodedSCG_Config := omit;
    var template (value) SRB_ToAddModList v_SRB_ToAddModList;
    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModList;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;
    var boolean v_IsFR1 := f_NR_CellInfo_GetIsFR1(nr_Cell1);
    var template (value) UL_AM_RLC v_ParametersTX := f_NR_UL_AM_RLC(v_IsFR1);
    var template (value) DL_AM_RLC v_ParametersRX := f_NR_DL_AM_RLC(v_IsFR1);
    
    // @siclog "Step 1" siclog@
    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertCiphAlg2Bitstring(f_NR_AS_CipheringAlgorithm_Get())), "NR Ciphering Alg"));
    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertIntAlg2Bitstring(f_NR_AS_IntegrityAlgorithm_Get())), "NR Integrity Alg"));
    //Set the NR ENDC Configuration for SRB1
    v_SRB_ToAddModList := {
      cs_NR_SRB_ToAddMod_ReestablishPDCP(tsc_SRB1, cs_NR_PDCP_Config_SplitSRB(-, f_NR_LogicalChannelId_SRB(tsc_SRB1),-,true)) 

    };
    v_RadioBearerConfig1 := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_RadioBearerConfig2 := cs_38508_RadioBearerConfigDef(v_SRB_ToAddModList, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));
    v_RLC_BearerToAddModList := {
      cs_38508_RLC_BearerConfig_SRB_ReestablishRLC(tsc_SRB1),
      cs_38508_RLC_BearerConfig_SRB(tsc_SRB3),
      cs_38508_RLC_BearerConfig_DRB(tsc_NR_DRB2,
                                    cs_38508_RLC_Config_AM_DRB(v_ParametersTX, v_ParametersRX),
                                    cs_38508_LogicalChannelConfig_DRB,
                                    true_)
    };
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, v_RLC_BearerToAddModList);
    //SS : configure SRB1 with RLCAckProhibit
    v_RadioBearerList := {
      cs_NR_SS_RadioBearer_Config(tsc_NR_RbId_SRB1, cs_NR_PDCP_Configuration_Proxy(crs_RlcBearerRouting_Eutra(eutra_Cell1)), cs_NR_RlcBearerConfig_AM(v_IsFR1, -, f_NR_LogicalChannelId_SRB(tsc_SRB1))),
      cs_NR_SS_RadioBearer_Config(tsc_NR_RbId_SRB3, cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsSRB))
    };
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_RadioBearerList, cs_TimingInfo_Now); //Configure SRB1 in NR with PDCP set to proxy + SRB3 on NR
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig1, v_SCGConfig, -, -, false);  // Now send octetstrings to EUTRA
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig2, , v_SCGConfig, -, -, false);      //Receive and activate KeNB keys
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);    
   // @siclog "Step 3" siclog@
   // SS transmits an NR RRCReconfiguration message to reconfigure SCG DRB on SRB3/NR Cell 1
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, -, cs_NR_MAC_CellGroupConfig(-, -, cs_NR_BSR_Config_Step3_822X1));
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config)));
    
    // @siclog "Step 4" siclog@
    // Check: Does the UE transmit an NR RRCReconfigurationComplete message on SRB3/NR Cell 1?
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 5" siclog@
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, { f_NR_LogicalChannelId_SRB(tsc_SRB3) });
    // Now send octetstrings to EUTRA
    f_NR_SendRRCReconfigurationContentsToEUTRA (cs_NR_RadioBearer_ReleaseSRB3, v_SCGConfig, -, -, false);

    // @siclog "Step 7-8" siclog@
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, { f_NR_LogicalChannelId_SRB(tsc_SRB1) });
    f_NR_SendRRCReconfigurationContentsToEUTRA (omit, v_SCGConfig, -, -, false); // Now send octetstrings to EUTRA

  }



After Change

function fl_TC_8_2_2_3_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC
  {// @sic R5-202647 sic@
    var template (value) CellGroupConfig v_SCGConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig1;
    var template (value) RadioBearerConfig v_RadioBearerConfig2;
    var template (omit) octetstring v_EncodedSCG_Config := omit;
    var template (value) SRB_ToAddModList v_SRB_ToAddModList;
    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModList;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;
    var boolean v_IsFR1 := f_NR_CellInfo_GetIsFR1(nr_Cell1);
    var template (value) UL_AM_RLC v_ParametersTX := f_NR_UL_AM_RLC(v_IsFR1);
    var template (value) DL_AM_RLC v_ParametersRX := f_NR_DL_AM_RLC(v_IsFR1);
    
    // @siclog "Step 1" siclog@
    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertCiphAlg2Bitstring(f_NR_AS_CipheringAlgorithm_Get())), "NR Ciphering Alg"));
    f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertIntAlg2Bitstring(f_NR_AS_IntegrityAlgorithm_Get())), "NR Integrity Alg"));
    //Set the NR ENDC Configuration for SRB1
    v_SRB_ToAddModList := {
      cs_NR_SRB_ToAddMod_ReestablishPDCP(tsc_SRB1, cs_NR_PDCP_Config_SplitSRB(-, f_NR_LogicalChannelId_SRB(tsc_SRB1),-,false)) , 
      cs_NR_SRB_ToAddMod_ReestablishPDCP(tsc_SRB2, cs_NR_PDCP_Config_SplitSRB(-, f_NR_LogicalChannelId_SRB(tsc_SRB2), -, false)) 
    };
    v_RadioBearerConfig1 := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_RadioBearerConfig2 := cs_38508_RadioBearerConfigDef(v_SRB_ToAddModList, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));
    v_RLC_BearerToAddModList := {
      cs_38508_RLC_BearerConfig_SRB_ReestablishRLC(tsc_SRB1),
      cs_38508_RLC_BearerConfig_SRB_ReestablishRLC(tsc_SRB2), 
      cs_38508_RLC_BearerConfig_SRB(tsc_SRB3),
      cs_38508_RLC_BearerConfig_DRB(tsc_NR_DRB2,
                                    cs_38508_RLC_Config_AM_DRB(v_ParametersTX, v_ParametersRX),
                                    cs_38508_LogicalChannelConfig_DRB,
                                    true_)
    };
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, v_RLC_BearerToAddModList);
    //SS : configure SRB1 with RLCAckProhibit
    v_RadioBearerList := {
      cs_NR_SS_RadioBearer_Config(tsc_NR_RbId_SRB1, cs_NR_PDCP_Configuration_Proxy(crs_RlcBearerRouting_Eutra(eutra_Cell1)), cs_NR_RlcBearerConfig_AM_SRB(f_NR_LogicalChannelId_SRB(tsc_SRB1))),
      cs_NR_SS_RadioBearer_Config(tsc_NR_RbId_SRB2, cs_NR_PDCP_Configuration_Proxy(crs_RlcBearerRouting_Eutra(eutra_Cell1)), cs_NR_RlcBearerConfig_AM_SRB(f_NR_LogicalChannelId_SRB(tsc_SRB2))),      cs_NR_SS_SRB3_Config
    };
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_RadioBearerList, cs_TimingInfo_Now); //Configure SRB1 in NR with PDCP set to proxy + SRB3 on NR
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig1, v_SCGConfig, -, -, false);  // Now send octetstrings to EUTRA
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig2, omit);  // Now send octetstrings to EUTRA 
    //Receive and activate KeNB keys
    f_ENDC_GetKeysActivateSecurity_KeNB(nr_Cell1, EUTRA, false); 
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);
    f_ENDC_GetKeysActivateSecurity(nr_Cell1, EUTRA, false);  
   // @siclog "Step 3" siclog@
   // SS transmits an NR RRCReconfiguration message to reconfigure SCG DRB on SRB3/NR Cell 1
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, -, cs_NR_MAC_CellGroupConfig(-, -, cs_NR_BSR_Config_Step3_822X1));
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config)));
    
    // @siclog "Step 4" siclog@
    // Check: Does the UE transmit an NR RRCReconfigurationComplete message on SRB3/NR Cell 1?
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // Send trigger to EUTRA
    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger); 
    //Receive and activate KeNB keys
    f_ENDC_GetKeysActivateSecurity_KeNB(nr_Cell1, EUTRA, false); 
    // @siclog "Step 5" siclog@
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, { f_NR_LogicalChannelId_SRB(tsc_SRB3) });
    // Now send octetstrings to EUTRA
    f_NR_SendRRCReconfigurationContentsToEUTRA (cs_NR_RadioBearer_ReleaseSRB3, v_SCGConfig, -, -, false);

    // @siclog "Step 7-8" siclog@
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, { f_NR_LogicalChannelId_SRB(tsc_SRB1) });
    f_NR_SendRRCReconfigurationContentsToEUTRA (omit, v_SCGConfig, -, -, false); // Now send octetstrings to EUTRA

  }


2.4 Change 4
	Function name
	f_ENDC_GetKeysActivateSecurity_KeNB

	Reason for change
	Default function used to activate security Keys f_ENDC_GetKeysActivateSecurity_KeNB assumes SRB1 PDCP SN need to be incremented by one. This is only valid for scenarios, in which SRB1 PDCP SN is not reset in the DL reconfig message. In test case 8.2.2.3.1, PDCP SN shall be reset after sending the DL reconfig message at step #1, hence PDCP SN should not be incremented by one, when the security keys are activated. New parameter boolean p_IncrSRB1  is introduced with default value true, which can be used in function := f_NR_RRC_CipherActTime_GetInitial to differentiate scenario in which SRB1 PDCp SN increment is not required.

	Summary of change
	Modified function f_ENDC_GetKeysActivateSecurity_KeNB to add New parameter boolean p_IncrSRB1 with default value true, which can be used in function := f_NR_RRC_CipherActTime_GetInitial() to differentiate scenario in which SRB1 PDCP SN increment is not required

	TTCN module
	NR_CommonFunctions_ENDC.ttcn

	MCC160 Comment
	



Before Change

function f_ENDC_GetKeysActivateSecurity_KeNB(NR_CellId_Type p_CellId,
                                               EUTRA_NR_CO_ORD_PORT p_Port ) runs on NR_BASE_PTC
  {
    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;
    var NR_SecurityParams_Type v_SecurityParams;
    
    v_SecurityParams := f_NR_Security_Get ();
    // get security from EUTRA
    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (p_Port);
    v_SecurityParams.SK_Counter := v_SecurityRx.SK_Counter;
    v_SecurityParams.S_KGNB := v_SecurityRx.Key;  //contains KeNB
    
    // get latest security keys
    v_SecurityParams := f_NR_InitAS_KeyChaining_KeNB (v_SecurityParams);
    f_NR_Security_Set(v_SecurityParams);
    
    // Configure PDCP security
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(p_CellId); 
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
                                          
  }



After Change

function f_ENDC_GetKeysActivateSecurity_KeNB(NR_CellId_Type p_CellId,
                                               EUTRA_NR_CO_ORD_PORT p_Port,
                                               boolean p_IncrSRB1 := true) runs on NR_BASE_PTC
  {
    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;
    var NR_SecurityParams_Type v_SecurityParams;
    
    v_SecurityParams := f_NR_Security_Get ();
    // get security from EUTRA
    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (p_Port);
    v_SecurityParams.SK_Counter := v_SecurityRx.SK_Counter;
    v_SecurityParams.S_KGNB := v_SecurityRx.Key;  //contains KeNB
    
    // get latest security keys
    v_SecurityParams := f_NR_InitAS_KeyChaining_KeNB (v_SecurityParams);
    f_NR_Security_Set(v_SecurityParams);
    
    // Configure PDCP security
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(p_CellId, p_IncrSRB1); 
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
                                          
  }


2.5 Change 5
	Function name
	f_ENDC_GetKeysActivateSecurity

	Reason for change
	Default function used to activate security Keys f_ENDC_GetKeysActivateSecurity assumes SRB1 PDCP SN need to be incremented by one. This is only valid for scenarios, in which SRB1 PDCP SN is not reset in the DL reconfig message. In test case 8.2.2.3.1, PDCP SN shall be reset after sending the DL reconfig message at step #1, hence PDCP SN should not be incremented by one, when the security keys are activated. New parameter boolean p_IncrSRB1  is introduced with default value true, which can be used in function := f_NR_RRC_CipherActTime_GetInitial() to differentiate scenario in which SRB1 PDCp SN increment is not required.

	Summary of change
	Modified function f_ENDC_GetKeysActivateSecurity to add New parameter boolean p_IncrSRB1 with default value true, which can be used in function := f_NR_RRC_CipherActTime_GetInitial() to differentiate scenario in which SRB1 PDCP SN increment is not required.

	TTCN module
	NR_CommonFunctions.ttcn

	MCC160 Comment
	



Before Change

function f_ENDC_GetKeysActivateSecurity(NR_CellId_Type p_CellId,
                                          EUTRA_NR_CO_ORD_PORT p_Port) runs on NR_BASE_PTC
  {
    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;
    var NR_SecurityParams_Type v_SecurityParams;
    
    v_SecurityParams := f_NR_Security_Get ();
    // get security from EUTRA
    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (p_Port);
    v_SecurityParams.SK_Counter := v_SecurityRx.SK_Counter;
    v_SecurityParams.S_KGNB := v_SecurityRx.Key; // @sic R5-194818 sic@
    
    // get latest security keys
    v_SecurityParams := f_NR_InitAS_KeyChaining_SKgNB (v_SecurityParams);
    f_NR_Security_Set(v_SecurityParams);
    
    // Configure PDCP security
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(p_CellId);  // @sic R5s190414 sic@ 
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
  }



After Change

function f_ENDC_GetKeysActivateSecurity(NR_CellId_Type p_CellId,
                                          EUTRA_NR_CO_ORD_PORT p_Port,
                                     boolean p_IncrSRB1 := true) runs on NR_BASE_PTC
  {
    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;
    var NR_SecurityParams_Type v_SecurityParams;
    
    v_SecurityParams := f_NR_Security_Get ();
    // get security from EUTRA
    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (p_Port);
    v_SecurityParams.SK_Counter := v_SecurityRx.SK_Counter;
    v_SecurityParams.S_KGNB := v_SecurityRx.Key; // @sic R5-194818 sic@
    
    // get latest security keys
    v_SecurityParams := f_NR_InitAS_KeyChaining_SKgNB (v_SecurityParams);
    f_NR_Security_Set(v_SecurityParams);
    
    // Configure PDCP security
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(p_CellId, p_IncrSRB1);  // @sic R5s190414 sic@ 
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
  }


2.6 Change 6
	Function name
	f_NR_RRC_CipherActTime_GetInitial

	Reason for change
	Default function used to activate security Keys f_NR_RRC_CipherActTime_GetInitial assumes SRB1 PDCP SN need to be incremented by one. This is only valid for scenarios, in which SRB1 PDCP SN is not reset in the DL reconfig message. In test case 8.2.2.3.1, PDCP SN shall be reset after sending the DL reconfig message at step #1, hence PDCP SN should not be incremented by one, when the security keys are activated. New parameter boolean p_IncrSRB1  is introduced with default value true, which can be used in function f_NR_RRC_CipherActTime_GetCurrent to differentiate scenario in which SRB1 PDCp SN increment is not required. 

	Summary of change
	New parameter boolean p_IncrSRB1  is introduced with default value true, which can be used in function f_NR_RRC_CipherActTime_GetCurrent to differentiate scenario in which SRB1 PDCp SN increment is not required

	TTCN module
	NR_SecuritySteps.ttcn

	MCC160 Comment
	



Before Change

function f_NR_RRC_CipherActTime_GetInitial(NR_CellId_Type p_CellId) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type
  {
    return f_NR_RRC_CipherActTime_GetCurrent(p_CellId); 
  }


After Change

function f_NR_RRC_CipherActTime_GetInitial(NR_CellId_Type p_CellId, boolean p_IncrSRB1 := true) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type
  {
    return f_NR_RRC_CipherActTime_GetCurrent(p_CellId, p_IncrSRB1); 
  }

2.7 Change 7
	Function name
	New template cs_NR_RlcBearerConfig_AM_SRB

	Reason for change
	Default template used to set up RLC for SRB f_NR_UL_AM_RLC & f_NR_DL_AM_RLC, configures the bearer with SnSize set as size18, which is incorrect. For SRB’s sn_FieldLength shall be size12. Used cs_38508_UL_RLC_SRB, cs_38508_DL_RLC_SRB to define parameters for SRB to set the IE sn_FieldLength to size12.

	Summary of change
	Defined new template cs_NR_RlcBearerConfig_AM_SRB to Use cs_38508_UL_RLC_SRB, cs_38508_DL_RLC_SRB to define parameters for SRB to set the IE sn_FieldLength to size12.

	TTCN module
	NR_ASP_Templates_RB_Config.ttcn

	MCC160 Comment
	



New template

template (value) NR_RlcBearerConfig_Type cs_NR_RlcBearerConfig_AM_SRB(template (omit) NR_LogicalChannelId_Type p_LogicalChannelId) :=
  /* @status    APPROVED (ENDC, NR5GC, NR5GC_IRAT) */
  //@sic R5s190109 change 8: added p_LogicalChannelId sic@
   cs_NR_RlcBearerConfig_Common(cs_NR_SS_RLC_Configuration_AM(cs_38508_UL_RLC_SRB, cs_38508_DL_RLC_SRB), p_LogicalChannelId); 





1. [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc51770681]Execution Log Files 
2. [bookmark: _Toc51770682]MediaTek MT6297
The MediaTek MT6297 5G NE UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_8_2_2_3_1_LOG.html
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
· PICS/PIXIT parameter file:
TC_8_2_2_3_1_PIXIT.xml

