
	Function name
	f_NG_EAP_SuccessfulAuthenticationResponse

	Reason for change
	As the EAP attributes are defined as a set, the order of the attributes included in the decoded EAP response sent to the TTCN will be listed in the order defined in the set, not in the order as they were received in the un-decoded message.  This can result in the MAC attribute being re-encoded in a different position in the EAP response.

	Summary of change
	Create a local function to find and clear the MAC value in the un-decoded message, rather than decoding and re-encoding the message

	TTCN module
	NG_NAS_SecurityFunctions



Before Change
	    function f_NG_EAP_SuccessfulAuthenticationResponse(inout NG_NAS_SecurityParams_Type p_Auth_Params,
                                                     octetstring p_ReceivedEAP,
                                                     template EAP_AT_RESULT_IND p_ResultInd := *,
                                                     O1_Type p_Id := '00'O)
    return boolean
  {
    var boolean v_Result := false;
    var bitstring v_ReceivedEAP := oct2bit(p_ReceivedEAP);
    var O16_Type v_RxdMAC;
    var O16_Type v_MACValue := '00000000000000000000000000000000'O;
    var bitstring v_K_aut;
    var EAP_Message_Type v_EAP_Message;
    var B32_128_Type v_RESValue;
    var bitstring v_EncodedMsg;
    var integer v_Len;
    var integer v_Paddinglen;
    var integer i;
    
    if (decvalue(v_ReceivedEAP, v_EAP_Message) != 0) {
      FatalError(__FILE__, __LINE__, "EAP message cannot be decoded");
    }
    
    //get length in bits of received RES (as it's a variable length, need to compare the correct length)
    v_Len := oct2int(v_EAP_Message.akaChallenge_Rsp.attributes.res.reslen);
    // store this in the global info
    p_Auth_Params.AuthParams.XRESLength := v_Len / 8;
    
    // Get the RES value from the global info
    v_RESValue := substr(p_Auth_Params.AuthParams.XRES, 0, v_Len);
    // Now pad it so that it's (hopefully) the same length of the received value
    if (v_Len mod 8 > 0 or (v_Len/8) mod 4 > 0) {
      v_Paddinglen := 8 - v_Len mod 8;   // pad to end of octet, in bits
      v_Paddinglen := (4 - ((v_Len + v_Paddinglen)/8) mod 4)*8; // pad to multiple of 4 octets, in bits
      for (i:=0; i < v_Paddinglen; i:=i+1) {
        v_RESValue := v_RESValue & '0'B;
      }
    }
    
    // Check message is AKAChallenge, length and AT_RES is correct
    if (match (v_EAP_Message, cr_EAP_Response_AKAChallenge(int2oct(lengthof(p_ReceivedEAP), 2), cr_EAP_AT_RES (bit2oct(v_RESValue), int2oct(v_Len, 2)), ?, p_ResultInd, p_Id))) { // @sic R5-201142, R5s201483 sic@
      // Then test MAC
      v_RxdMAC := v_EAP_Message.akaChallenge_Rsp.attributes.mac.mac;
      
      // First build the message with MAC=0
      v_EAP_Message.akaChallenge_Rsp.attributes.mac.mac := v_MACValue;
      // Compute MAC according to RFC 5448 cl. 3.4.2
      v_EncodedMsg := encvalue(v_EAP_Message);
      v_K_aut := substr(p_Auth_Params.MK, 128, 256); // RFC 5448 cl. 3.3 K_aut to be used for computing MAC @sic R5s200400 sic@
      v_MACValue := bit2oct(substr(f_HMAC_SHA_256(v_K_aut, bit2oct(v_EncodedMsg)), 0, 128)); // @sic R5s200400 sic@
      // Then test it against the received value
      if (match(v_RxdMAC, v_MACValue)) { // if the MAC also matches then the whole message is correct
        v_Result := true;
      }
    }
    return v_Result;
  }



After Change
	  function fl_NG_EAP_Response_ClearMACValue(octetstring p_ReceivedEAP) return octetstring
  {
    var octetstring v_ModifiedEAP := p_ReceivedEAP;
    var integer i := 8;                         /* according to RFC 4187 clause 8.1 the EAP-AKA header is 8 octets */
    var integer k;
    
    while (p_ReceivedEAP[i] != '0B'O) {         /* MAC attribute type is 11 according to RFC 4187 clause 11 */
      i := i + (oct2int(p_ReceivedEAP[i+1]) * 4);
      if (i > lengthof(p_ReceivedEAP)) {
        FatalError(__FILE__, __LINE__, "no MAC attribute found");
      }
    }
    i := i + 4;                                 /* first 4 octets are type, length, reserved (RFC 4187 clause 10.15) */
    for (k := 0; k < 16; k := k + 1) {
      v_ModifiedEAP[i+k] := '00'O;
    }
    return v_ModifiedEAP;
  }

  function f_NG_EAP_SuccessfulAuthenticationResponse(inout NG_NAS_SecurityParams_Type p_Auth_Params,
                                                     octetstring p_ReceivedEAP,
                                                     template EAP_AT_RESULT_IND p_ResultInd := *,
                                                     O1_Type p_Id := '00'O)
    return boolean
  {
    var boolean v_Result := false;
    var bitstring v_ReceivedEAP := oct2bit(p_ReceivedEAP);
    var O16_Type v_RxdMAC;
    var O16_Type v_MACValue := '00000000000000000000000000000000'O;
    var bitstring v_K_aut;
    var EAP_Message_Type v_EAP_Message;
    var B32_128_Type v_RESValue;
    var octetstring v_ModifiedEAP;
    var bitstring v_EncodedMsg;
    var integer v_Len;
    var integer v_Paddinglen;
    var integer i;
    
    if (decvalue(v_ReceivedEAP, v_EAP_Message) != 0) {
      FatalError(__FILE__, __LINE__, "EAP message cannot be decoded");
    }
    
    //get length in bits of received RES (as it's a variable length, need to compare the correct length)
    v_Len := oct2int(v_EAP_Message.akaChallenge_Rsp.attributes.res.reslen);
    // store this in the global info
    p_Auth_Params.AuthParams.XRESLength := v_Len / 8;
    
    // Get the RES value from the global info
    v_RESValue := substr(p_Auth_Params.AuthParams.XRES, 0, v_Len);
    // Now pad it so that it's (hopefully) the same length of the received value
    if (v_Len mod 8 > 0 or (v_Len/8) mod 4 > 0) {
      v_Paddinglen := 8 - v_Len mod 8;   // pad to end of octet, in bits
      v_Paddinglen := (4 - ((v_Len + v_Paddinglen)/8) mod 4)*8; // pad to multiple of 4 octets, in bits
      for (i:=0; i < v_Paddinglen; i:=i+1) {
        v_RESValue := v_RESValue & '0'B;
      }
    }
    
    // Check message is AKAChallenge, length and AT_RES is correct
    if (match (v_EAP_Message, cr_EAP_Response_AKAChallenge(int2oct(lengthof(p_ReceivedEAP), 2), cr_EAP_AT_RES (bit2oct(v_RESValue), int2oct(v_Len, 2)), ?, p_ResultInd, p_Id))) { // @sic R5-201142, R5s201483 sic@
      // Then test MAC
      v_RxdMAC := v_EAP_Message.akaChallenge_Rsp.attributes.mac.mac;
      
      // First build the message with MAC=0
      v_ModifiedEAP := fl_NG_EAP_Response_ClearMACValue(p_ReceivedEAP);
      v_EAP_Message.akaChallenge_Rsp.attributes.mac.mac := v_MACValue;
      // Compute MAC according to RFC 5448 cl. 3.4.2
      v_EncodedMsg := encvalue(v_EAP_Message);
      v_K_aut := substr(p_Auth_Params.MK, 128, 256); // RFC 5448 cl. 3.3 K_aut to be used for computing MAC @sic R5s200400 sic@
      v_MACValue := bit2oct(substr(f_HMAC_SHA_256(v_K_aut, bit2oct(v_ModifiedEAPv_EncodedMsg)), 0, 128)); // @sic R5s200400 sic@
      // Then test it against the received value
      if (match(v_RxdMAC, v_MACValue)) { // if the MAC also matches then the whole message is correct
        v_Result := true;
      }
    }
    return v_Result;
  }



